Logisch Programmeren en Zoektechnieken

Ulle Endriss
Institute for Logic, Language and Computation
University of Amsterdam

[https://staff.science.uva.nl/u.endriss/teaching/prolog/](https://staff.science.uva.nl/u.endriss/teaching/prolog/)
Logic and Prolog

This final part of the course will be dedicated to the connections between logic and Prolog. This is a two-way street:

- **Logical Foundations of Prolog:**
  Prolog programs can be interpreted as sets of formulas in first-order logic. And when processing a query, Prolog is applying the rules of a logical proof system called *resolution*.

- **Applications of Prolog to Logic:**
  Prolog is a programming language that is particularly well suited to solving problems involving logic. We will see how to implement an *automated theorem prover* based on another proof system, *semantic tableaux*.
# Correspondences

<table>
<thead>
<tr>
<th>Prolog</th>
<th>First-order Logic (FOL)</th>
</tr>
</thead>
<tbody>
<tr>
<td>predicate</td>
<td>predicate</td>
</tr>
<tr>
<td>argument</td>
<td>term</td>
</tr>
<tr>
<td>variable</td>
<td>universally quantified variable</td>
</tr>
<tr>
<td>atom</td>
<td>constant/function/predicate symbol</td>
</tr>
<tr>
<td>sequence of subgoals</td>
<td>conjunction</td>
</tr>
<tr>
<td>:-</td>
<td>implication (other way round)</td>
</tr>
</tbody>
</table>
Question

What is the logical meaning of this program?

```prolog
bigger(elephant, horse).
bigger(horse, donkey).
is_bigger(X, Y) :- bigger(X, Y).
is_bigger(X, Y) :- bigger(X, Z), is_bigger(Z, Y).
```
Answer

The translation of a Prolog program is usually represented as a set of formulas, with each formula corresponding to one of the clauses in the program:

\[
\{ \begin{align*}
&\text{bigger(elephant, horse),} \\
&\text{bigger(horse, donkey),} \\
&\forall x.\forall y.(\text{bigger}(x,y) \rightarrow \text{is\_bigger}(x,y)), \\
&\forall x.\forall y.\forall z.(\text{bigger}(x,z) \land \text{is\_bigger}(z,y) \rightarrow \text{is\_bigger}(x,y)) 
\end{align*} \}
\]

Such a set is to be interpreted as the conjunction of all the formulas in the set.
Translation of Programs

- Predicates remain the same (syntactically).
- Commas separating subgoals become $\land$.
- $\leftarrow$ becomes $\rightarrow$ and the order of head and body is changed.
- Every variable is bound by a universal quantifier ($\forall$), with the scope of the quantifier being the full implication.
Meaning of Prolog Programs

A Prolog program corresponds to a set of formulas, all of which are taken to be true. This restricts the range of possible interpretations of the predicate and function symbols appearing in these formulas. The formulas in the translated program may be thought of as the premises in a proof.

If Prolog gives a positive answer to a given query, this means that the translation of the query is a logical consequence of these premises (possibly under the assumption of suitable variable instantiations). If Prolog gives a negative answer, this means that the query cannot be shown to be true under the assumption that all of the program formulas are true.
Translation of Queries

Queries are translated like rules; the “empty head” is translated as $\bot$ (falsum). This corresponds to the negation of the goal whose provability we are trying to test when submitting a query to Prolog:

$$Goal \rightarrow \bot$$

is logically equivalent to $\neg Goal$

Thus, instead of deriving the goal directly, we try to prove that adding the negated goal to the program will lead to a contradiction:

$$Premises \models Goal \quad \text{if and only if} \quad Premises, (Goal \rightarrow \bot) \models \bot$$
Example

The query

?= is_bigger(elephant, X), is_bigger(X, donkey).

corresponds to the following first-order formula:

\( \forall x. (is\_bigger(elephant, x) \land is\_bigger(x, donkey) \rightarrow \bot) \)

Remark: Intuitively, a query such as the one above asks “is there an \( X \) such that …”. To see that our systematic translation above is correct, note that \( \neg(\exists x. P(x)) \equiv \forall x. (\neg P(x)) \equiv \forall x. (P(x) \rightarrow \bot). \)
Horn Formulas

The (matrices of the) formulas we get when translating from Prolog all have the same structure:

\[ A_1 \land A_2 \land \cdots \land A_n \rightarrow B \]

Such a formula can be rewritten as follows:

\[ A_1 \land A_2 \land \cdots \land A_n \rightarrow B \equiv \]
\[ \neg(A_1 \land A_2 \land \cdots \land A_n) \lor B \equiv \]
\[ \neg A_1 \lor \neg A_2 \lor \cdots \lor \neg A_n \lor B \]

Hence, formulas obtained from translating Prolog formulas can always be rewritten as disjunctions of literals with at most one positive literal. Such formulas are know as Horn formulas.
Prolog and Resolution

The search tree built up by Prolog when trying to answer a query corresponds to a logic proof using resolution, which is a very efficient proof system for Horn formulas.

A short introduction can be found in the Lecture Notes; for more details refer to theoretically oriented books on logic programming.

Here we are only going to introduce resolution for propositional logic (corresponding to Prolog without variables) . . .

Remark: In principle, any other proof system could be used as well (e.g., tableaux), but historically Prolog is based on resolution.
Propositional Resolution

Problem: The purpose of the resolution procedure is to show that a given set of formulas is unsatisfiable.

Why is that (proving unsatisfiability) an interesting problem?

Answer: \( \mathcal{P} \cup \{\neg G\} \) is unsatisfiable \( \text{iff} \) \( G \) follows from the set \( \mathcal{P} \).

Normal form: Recall that any formula can be transformed into an equivalent formula in conjunctive normal form (CNF). Also, a conjunction is satisfiable \( \text{iff} \) the set of its conjuncts is. Hence, it suffices to consider only sets of disjunctions of literals . . .

Remark: In general, translation into CNF is not a trivial matter (e.g., your formula might get exponentially longer). But translated Prolog clauses already are in CNF, so we are fine here.
The Resolution Rule

Resolution for propositional logic consists of a single rule, which we apply to pairs of disjunctions until we find an explicit contradiction.

To simplify presentation, we write disjunctions as sets of literals. Example: $A \lor \neg B$ becomes \{A, \neg B\}.

The resolution rule for propositional logic (and an example):

\[
\begin{align*}
\text{Dis}_1 \cup \{L\} & \quad \{A, \neg B, C, D\} \\
\text{Dis}_2 \cup \{\neg L\} & \quad \{\neg C, D, \neg E\} \\
\hline
\text{Dis}_1 \cup \text{Dis}_2 & \quad \{A, \neg B, D, \neg E\}
\end{align*}
\]

Make sure you understand why this is a sound rule . . .

The simplification of $D \lor D$ to $D$ is implicit (a.k.a. factoring).

An explicit contradiction is reached when both $\text{Dis}_1$ and $\text{Dis}_2$ are empty. In this case, we usually write $\square$ (or $\bot$) for the resolvent.
Examples

• Prove that $P$ and $P \rightarrow Q$ entail $Q$ (first translate!).

• Prove that the following set of formulas is unsatisfiable:

$$\{ A \lor B, A \lor \neg B, \neg A \lor B, \neg A \lor \neg B \}$$

• Consider the following Prolog program:

```
happy :- summer, france.
happy :- rich.
rich :- lottery.
summer.
lottery.
```

Simulate how Prolog would react to the following query by giving the appropriate resolution proof:

?- happy.
**SLD Resolution**

If all formulas are Horn formulas, then we can apply resolution in a “systematic” manner (called SLD resolution) and be sure that we always find a proof if there exists one at all. Intuitively, this is why Prolog “works” (reasoning in general FOL would be much harder).

SLD stands for **Selective Linear Resolution for Definite clauses**:

- **Linearity**: we start with the only negative Horn formula (the query) and then always use the previous resolvent (new query).

- The *selection function* is very simple: it always chooses the first literal (in the current “query”).

- The input is restricted to one negative Horn formula (query) and a number of *positive Horn formulas* (rules and facts). Positive Horn formulas are also known as “definite clauses”.

The above still leaves open which matching literal to choose; in practice this issue is dealt with via *backtracking*. 
Matching vs. Sound Unification

Prolog’s matching algorithm does not implement the so-called \textit{occurs-check}. This is a good thing, because Prolog would be very inefficient otherwise, but special care is required in the rare cases where sound unification is important.

This is how SWI-Prolog reacts to a query where this matters:

\begin{verbatim}
?- X = f(X).
X = f(X)
Yes
\end{verbatim}

It used to be something like this, which makes things much clearer:

\begin{verbatim}
?- X = f(X).
X = f(f(f(f(f(f(f(f(f(f(f(f(...)))))))))))
Yes
\end{verbatim}

But note that this is still \textit{wrong}, if you want real unification . . .
Sound Unification in SWI-Prolog

Fortunately, SWI-Prolog comes with a built-in predicate for sound unification that we can use. Examples:

\[
?- \text{unify\_with\_occurs\_check}(X, f(X)).
\]
No

\[
?- \text{unify\_with\_occurs\_check}(X, f(Y)).
\]
\[X = f(Y)\]
Yes

This predicate performs matching of the two arguments but also checks that the first does not occur (strictly) within the second.
Summary: Logic Foundations

• Prolog programs (without advanced features) correspond to sets of first-order logic (Horn) formulas.

• During translation, \( \text{:-} \) becomes an implication (from right to left), commas between subgoals become conjunctions, and all variables need to be universally quantified. Queries become (universally quantified) implications with \( \perp \) in the consequent.

• Prolog’s search to satisfy a query corresponds to a logical proof. In principle, any proof system could be used. Historically, Prolog is based on resolution, which is particularly suited as it is tailored to Horn formulas.

• For efficiency reasons, Prolog uses a simplified matching algorithm rather than full sound unification.
Automated Reasoning

We now want to implement a theorem prover in Prolog: a program that can check automatically whether a given set of formulas (the *premises*) logically entail a further formula (the *conclusion*).

We will restrict ourselves to *classical propositional logic* (but people have developed provers for many other logics as well).

Our prover will be based on the method of *semantic tableaux*.
Semantic Tableaux

Aim: Show that a given set of formulas is unsatisfiable (if you can do this, you can also do logical consequence or tautology checking).

Procedure: Start a tableau (a binary tree) with those formulas in the root. Apply elimination rules to formulas, resulting in new formulas being added to the end of a branch, and possibly in branches being split in two. Close a branch when it includes an obvious contradiction. Succeed if you can close all branches.

People use different ways of representing the formulas on the tableau. Some distinguish between formulas claimed to be true and those claimed to be false. You probably know this version:

| formulas claimed to be true | formulas claimed to be false |

For our implementation, it will be easier to explicitly label each formula as being either (claimed to be) true or false.
Elimination Rules and Branch Closure

<table>
<thead>
<tr>
<th>Negation</th>
<th>Conjunction</th>
<th>Disjunction</th>
<th>Closure</th>
</tr>
</thead>
<tbody>
<tr>
<td>( t :: \neg A )</td>
<td>( t :: A \land B )</td>
<td>( t :: A \lor B )</td>
<td></td>
</tr>
<tr>
<td>( f :: A )</td>
<td>( t :: A )</td>
<td>( t :: A \mid t :: B )</td>
<td>( t :: A )</td>
</tr>
<tr>
<td>( f :: \neg A )</td>
<td>( t :: B )</td>
<td></td>
<td>( f :: A \lor B )</td>
</tr>
<tr>
<td>( t :: A )</td>
<td>( f :: A \mid f :: B )</td>
<td>( f :: A )</td>
<td>( f :: B )</td>
</tr>
<tr>
<td></td>
<td></td>
<td>( f :: A \times )</td>
<td></td>
</tr>
</tbody>
</table>
Examples

Give tableau proofs for the following claims:

(1) \( p \lor \neg p \) is a tautology

(2) \( \{ p \lor q, p \lor \neg q, \neg p \} \) is unsatisfiable

(3) \( \neg p \lor \neg q \models \neg(p \land q) \) [logical consequence]
Semantic Tableaux in Prolog

In 1995, Bernhard Beckert and Joachim Posegga (Uni Karlsruhe) published a paper in the *Journal of Automated Reasoning* in which they showed how to implement a sound, complete, and efficient theorem prover for first-order logic based on semantic tableaux in Prolog requiring a total of 360 (!) characters (including spaces):

\[
\begin{align*}
\text{prove}((E,F),A,B,C,D) & : - !, \text{prove}(E,[F|A],B,C,D). \\
\text{prove}((E;F),A,B,C,D) & : - !, \text{prove}(E,A,B,C,D), \text{prove}(F,A,B,C,D).
\end{align*}
\]

\[
\begin{align*}
\text{prove}(\text{all}(I,J),A,B,C,D) & : - !, \\
& \neg \text{length}(C,D), \text{copy_term}((I,J,C),(G,F,C)), \\
& \text{append}(A,[\text{all}(I,J)],E), \text{prove}(F,E,B,[G|C],D).
\end{align*}
\]

\[
\begin{align*}
\text{prove}(A,_,[C|D],_,_,_) & : - \\
& ((A=-(B);-(A)=B) \rightarrow (\text{unify}(B,C);\text{prove}(A,[]),D,_,_,_))). \\
\text{prove}(A,[E|F],B,C,D) & : - \text{prove}(E,F,[A|B],C,D).
\end{align*}
\]

We will not try to understand this. But we will implement a prover for propositional logic using a very similar approach.
Representing Formulas

We need operators for the basic connectives of propositional logic:

\[
\text{:- op}(100, \text{fy}, \text{neg}), \\
\text{ op}(200, \text{yfx}, \text{and}), \\
\text{ op}(300, \text{yfx}, \text{or}).
\]

We also declare an operator that we will use to label formulas:

\[
\text{:- op}(600, \text{xfx}, ::).
\]

Examples:

\[
p \land \lnot q \text{ claimed to be } true \quad \rightarrow \quad t :: p \text{ and } \lnot q
\]

\[
\lnot (p \lor q) \text{ claimed to be } false \quad \rightarrow \quad f :: \lnot (p \text{ or } q)
\]
Representing Branches

A branch in a tableau is the list of formulas we find when we go from the root to one of the leaf nodes. We will model each branch in terms of two lists:

- **Forms**: the labelled non-atomic formulas (still) on the branch
- **Atoms**: the labelled atomic formulas on the branch

As long as there are still non-atomic formulas in **Forms**, we can continue to apply elimination rules. After we have applied a rule to a given formula, we remove that formula from **Forms**.

Eventually, only labelled atoms will be left (**Forms** will be empty).
Grand Plan

We will implement a predicate with this signature:

\[
\text{tableau}(\text{+Forms}, \text{+Atoms})
\]

It should succeed if a tableau starting from a branch with the labelled complex formulas in \( \text{Forms} \) and the labelled atomic formulas in \( \text{Atoms} \) can eventually be closed (using our rules).

For example, to verify that \( p \lor \neg p \) is a tautology we initialise the tableau with \( p \lor \neg p \) (labelled as \textit{false} \) and an empty list of atoms:

\begin{verbatim}
?- tableau([f :: p or neg p], []).
Yes
\end{verbatim}

That is, assuming that \( p \lor \neg p \) is false allowed us to successfully close the tableau, i.e., to derive a contradiction.
Closing Branches

We can close a branch when we find the same formula on it twice, once labelled as *true* and once labelled as *false*.

We can restrict ourselves to closing branches using contradictorily labelled *atoms* only (think about this, it is not obvious).

- Closing only on atoms is a good strategy for a computer:
  Checking to see whether we can close a branch is the most costly operation (*why?*), so we don’t want to do this too often.

- Closing on arbitrary formulas is a better strategy for humans:
  As a human, you want to build as small a tableau as you can.

Our rule for closing branches (base case of the recursion):

```prolog
tableau(_, Atoms) :- % don’t need to know Forms
    member(t :: A, Atoms),
    member(f :: A, Atoms), !. % no need to backtrack
```
Elimination Rules: Negation

We translate each tableau elimination rule into a Prolog rule. The idea is always the same:

1. check whether the head of Forms is of a given type;
2. apply the correct elimination rule to the head;
3. continue proof search with the head being replaced by the new formula (this will be slightly more complex for branching rules).

Here are the two elimination rules for negation:

\[
\text{tableau}([t :: \neg A \mid \text{Forms}], \text{Atoms}) :- !, \\
\text{tableau}([f :: A \mid \text{Forms}], \text{Atoms}).
\]

\[
\text{tableau}([f :: \neg A \mid \text{Forms}], \text{Atoms}) :- !, \\
\text{tableau}([t :: A \mid \text{Forms}], \text{Atoms}).
\]

The list of Atoms doesn’t change. The cut is applied once we know that this is the rule to use (after that we never want to backtrack).
Eliminating True Conjunctions

If the first formula on the branch is a conjunction (labelled true), then replace it with its two conjuncts (both labelled true):

\[
\text{tableau}([t :: A \text{ and } B \mid \text{Forms}], \text{Atoms}) :- !, \\
\text{tableau}([t :: A, t :: B \mid \text{Forms}], \text{Atoms}).
\]
Branching: Eliminating True Disjunctions

To close a branch with a (true) disjunctive formula on it, we need to split that branch and close the two resulting branches:

\[
\text{tableau([t :: A or B | Forms], Atoms)} \leftarrow !, \\
\text{tableau([t :: A | Forms], Atoms),} \\
\text{tableau([t :: B | Forms], Atoms).}
\]

Careful! We are not using disjunction (\(\lor\)) in the body of our rule. We need to show that the left branch will close eventually and that the right branch will close eventually.
Eliminating False Conjunctions and Disjunctions

These work exactly as the other rules we have seen:

```prolog
tableau([f :: A and B | Forms], Atoms) :- !,
    tableau([f :: A | Forms], Atoms),
    tableau([f :: B | Forms], Atoms).

tableau([f :: A or B | Forms], Atoms) :- !,
    tableau([f :: A, f :: B | Forms], Atoms).
```
**Final Step: Moving Atoms**

We need one more rule. Whenever we encounter an atomic formula in `Forms` we need to move it to `Atoms`:

```
tableau([Label :: Atom | Forms], Atoms) :-
   atom(Atom), !,
   tableau(Forms, [Label :: Atom | Atoms]).
```

That’s it! And it works:

```
?- tableau([t :: p, t :: neg p or q, f :: q], []).
Yes

?- tableau([t :: p or neg p], []).  
No
```

Note that the order of the rules of the program does not matter: any order will lead to a correct program. But the order might affect efficiency. Can you predict what would be a good order?
User Interface: Tautology Checking

The following predicate takes a formula and then appropriately initialises a tableau to check whether it is a tautology:

\[
tautology(\text{Formula}) :- \\
\text{tableau}([f :: \text{Formula}], []). \\
\]

Examples:

?- tautology(p or neg p).
Yes

?- tautology(neg (neg (p or (q and r)) and p) or q).
Yes
Summary: Automated Reasoning

This has been a case study in automated reasoning:

- We have seen how to implement a tableau-based theorem prover in Prolog. The main data structure is that of lists of formulas (representing branches) and elimination rules are implemented as operations on the heads of those lists.

- Despite its simplicity, our program will perform rather well also on larger examples.

- Note that our implementation only checks whether a closed tableau exists for a given set of formulas—it does not return that tableau to the user! Of course, you can also write a program for that task, but it is a bit more difficult.

Automated reasoning is an important topic in AI and people have developed all sorts of automated reasoners for different types of problems, modelled in different logics as well as other formalisms.